
1. COORDINATE SYSTEM 

 

The default PPLZ coordinate system is defined in Figure 1-1. 

 

 

   (0,0)    +x 

     +y 

    

 

 

 

 

          

            

             

 Label     

 Feed      

 Direction (exit)  

   

Fig. 1-1 PPLZ coordinate system 

 

The default origin point (0,0) is at the top left point of a label. Negative 

coordinate values are not allowed. The origin point can be changed to 

right bottom corner under ^POI mode. The ranges of X and Y coordinates 

are: 

 

 Minimum Maximum 

X coordinate 0 
About 4 inches. The actual value 

depends on the printer model. 

Y coordinate 0 
About 48 inches. The actual value 

depends on the printer model. 

 

Orientation 

There are four orientations. You may rotate the pattern for readability or 

other special purposes. 

 

-Portrait 

-Landscape 

-Reverse portrait 

-Reverse landscape 

    

Fig. 1-2 Four orientations 

2. COMMAND FORMATS 

 

The PPLZ command set is a software interface that enables the 

communication between the printer and host. By that the printer prints 

the exact patterns for the host and user. 

 
 

Basic Syntax 

 

The general command consists of a prefix character, two-character 

mnemonic code and a parameter string.  

 

<leading code> <2-character command> [parameters] 

 

All characters except <CR> and <LF> are printable ones, not control codes. 

The two characters for command type must be upper-case capitals. 

Some special codes for special-purpose function can be redefined. These 

include: 

 

Code  Name  Function                   

^  caret  used as a leading character 

~  tilde   used as a leading character 

,  delimiter  separator between parameters 

 

The line separator (<CR><LF>, <CR> or <LF>) is used just for human 

readability and ignored by the printer. The space codes can be inserted 

between commands. 

Once the parameters in the command are omitted the default settings or 

parameters saved in the E2PROM or flash will be used. 

 

 

Command Categories 

 

All the PPLZ commands fall into several categories as following: 

-Format package 

-Label definition 

-Field default 

-Format default 

-Format rotation 

-Printer control 

-Alphanumeric field 

-Bar code 

-Text or font 

-Graphic image 

File name 

 

The objects like soft fonts, graphics or forms can be accessed through 

their file names. The full file name same as that under MS-DOS consists 

of  

-One character for device 

-Max. 8-character file name 

-Max. 3-character extension name (optional) 

 

For example, 

 R:ABC.GRF 

 

Device name 

R: for default RAM , B: for USB memory and E: for Flash memory. 

 

Extension name 

The default extension name for the graphic is “ .GRF ”, “ .FNT ” is for 

the soft font and “ .PPLZ” is for the form. 

 

 

Skeleton 

 

Commands with a leading code, “ ~ “ can be independently interpreted. 

Others must be in the following package. 

 

  ^XA 

  …. 

  ^XZ 

Settings 

 

The settings affect either the printout pattern or the control to the 

printer. For example, ^PO command defines the label rotation. Some 

settings are stored in the flash or E2PROM. E2PROM is a hardware device 

and can permanently save the parameters. While the others are set to 

fixed default values each time the printer is started. They can be changed 

by related commands. 

3. FONTS 

 

The fonts can be classified into some categories by their characteristics. 

 

Residence 

Internal fonts or downloadable soft fonts. The internal fonts are built in 

the printer. They are fixed and cannot be erased or cleared. The soft 

fonts are from the host. Users can download the soft fonts from the host 

by their needs and erased them by proper commands to avoid memory 

full. 

 

Scalability 

Scalable fonts and bitmap fonts. All fonts can be expanded but only 

scalable fonts can expand smoothly. The zigzag phenomenon occurs for 

the bitmap fonts when scaling. 

 

Spacing 

Fixed and proportional fonts. The spacing of fixed font is identical for all 

characters while it depends on the characters for proportional font. 

 

For example: 

 

   Fixed font 

   Proportional font 

TrueType Fonts 

 

The format of scalable fonts is the so-called TrueType. Since it is a 

Windows standard it can make your aication WYSIWYG, What You See Is 

What You Get. 

 

 

 

Internal Fonts 

 

There are 15 internal bitmap fonts and one scalable font. Their related 

data are tabled as follow. 

Font ID 
Matrix 

(width x height) 
Type Remark 

A 5 x 9 ULD  

B 7 x 11 U  

C, D 10 x 18 ULD 
C & D are 

identical. 

E 15 x 28 OCR B  

F 13 x 26 ULD  

G 40 x 60 ULD  

H 13 x 21 OCR A  

GS 24 x 24 SYMBOL  

0 Default: 12 x 15 
CG Triumvirate 

Bold Condensed 
Scalable font 

P 18 x 20 ULD  

Q 24 x 28 ULD  

R 31 x 35 ULD  

S 35 x 40 ULD  

T 42 x 48 ULD  

U 53 x 59 ULD  

V 71 x 80 ULD  

 

 

Symbol Sets 

 

The scalable font (CG Triumvirate Bold Condensed) supports PC850 

symbol set while other bitmap fonts support several international symbol 

sets. Refer to the ^CI command for details. 

Point Size 

 

The point size is a rough length measurement.  

1 point = 1/72 inch = 2.8 pixels for character cell height under a 203 

DPI printer 

4. FORMS 

 

If you need lots of printouts that are almost same except some fields, 

you may just send those different field data from the second set of label. 

This can save huge data and communication time. 

 

A form may consist of 

-Common fields or patterns. 

-Variable data fields. 

-Counter fields, which can automatically increase or decrease. 

 
 

Procedure 

 

To use the form you must follow the procedure as below. 

 

1. The form defined by ^DF command. 

In this you also define the related fields like counter. 

2. The form retrieval. 

By ^XF and specifying the form name to execute the form. 

3. Initial values for counters and variables. 

4. Repeat steps 2 and 3 for next set of labels. 

 

Refer to the relevant commands for details. 
 

 

Example 

Form definition ^XA 

^DFR:FORM.PPLZ^FS 

^FO50,60^AD^FN1^FS 

^FO50,110^AD^FN2^FS 

^FO50,10^AF^FDFIXED DATA^FS 

^XZ 

Form execution I ^XA^XFR:FORM.PPLZ^FS 

^FN1^FDVAR 1^FS 

^FN2^FDVAR 2^FS 

^XZ 

Form execution II ^XA^XFR:FORM.PPLZ^FS 

^FN1^FDDATA1^FS 

^FN2^FDDATA2^FS 

^XZ 

 

Output 

5. COMMANDS 

 

In this section we list all commands by alphabetical order with 

description like 

 

-Command format. 

-Parameter definition and default value. 

-Related data structures. 

-Related commands. 

-Example. 

 

For most commands they must be combined together with others in 

order to print specific patterns or perform special function. 

 

“FNT” format of PPLZ (Zebra) is not compatible.  

(“FNT” is the extension name of Bitmap font.) 

^A Font Selection 

 

Syntax  ^Afo,h,w 

 

Parameter  

f Font ID 

Default: 0 (Internal scalable font, CG Triumvirate Bold 

Condensed) 

Others: 0 ~ 9, A ~ Z 

The ID may be defined through the ^CW command. 

The ID, ‘:’, ‘;’ or ‘>’ are reserved for 2 byte fonts, like Chinese, 

Taiwanese, Korean, etc.. 

o Orientation 

Default: N (Portrait) 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Cell height in pixels 

For bitmap font the h parameter is used to calculate the 

times to scale the character height.  

vsf (vertical scale factor) = (h+ch/2)/ch 

Where ch is the original cell height. The vsf must be an 

integer and its minimum value is 1. 

w Cell width in pixels 

For bitmap font the h parameter is used to calculate the 

times to scale the character width. 

hsf (hertical scale factor) = (w+cw/2)/cw 

Where cw is the original cell width. The hsf must be an 

integer and its minimum value is 1. 

Example   

 

^XA 

^FO20,20^A0,25,15^FDScalabe Font^FS 

^FO30,70^AA^FDFONT A^FS 

^FO150,70^AA,20,18^FDFONT A^FS 

^FO30,90^AB^FDFONT B^FS 

^FO30,120^AC^FDFONT C^FS 

^XZ 

 

Output 

 

 

Example 
 

The following example prints the Taiwanese characters if the Taiwanese 

font card is installed. 

 

^XA 

^FO100,120^A:,48,96^FD中文測試^FS 

^XZ 

 

Output 

^A@ Font Selection by the font name 

 

Syntax  ^A@o,h,w,n 

 

Parameter 

o Orientation 

Default: N (Portrait) 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Cell height in pixels 

For bitmap font the h parameter is used to calculate the 

times to scale the character height. 

vsf (vertical scale factor) = (h+ch/2)/ch 

Where ch is the original cell height. The vsf must be an 

integer and its minimum value is 1. 

w Cell width in pixels 

For bitmap font the h parameter is used to calculate the 

times to scale the character width.  

hsf (hertical scale factor) = (w+cw/2)/cw 

Where cw is the original cell width. The hsf must be an 

integer and its minimum value is 1. 

n An optional device character. A max 8-chacter file name and 

a max 3-character extension name. 

e.g.  R:ABC.FNT 

 

Example 

 

mailto:%5eA@o,h,w,n

^XA 

^FO20,50^A@,25,22,B:COMIC.FNT^FDThis Is A Soft 

Font^FS 

^XZ 

 

Output 

^B0 Bar code – Aztec 

 

Syntax  ^B0a,b,c,d,e 

 

Parameter 

a Orientation 

Default: N (Portrait) 

b Magnification factor 

Default: 2 for 200 dpi printers; 3 for 300 dpi printers. 

Others: 1~10. 

c ECIC (Extended channel interpretation code) indicator 

Default: N (Data includes ECIC) 

Y (Data does not include ECIC) 

d Error correction level & Symbol size/barcode type indicator 

Default: 0 (error correction level) 

Others: 01~99: error correction percentage 

101~104: 1~4-layer compact symbol 

201~232: 1~32-layer full-range symbol 

300: a simple Aztec “Rune” 

e Menu symbol indicator 

Default: N is for it to be a menu symbol. 

Y is not for it to be a menu symbol. 

 

Example 

^XA 

^FO20,250^B0N,6,N,0,N^FDAztec barcode test^FS  

^XZ 

 

Output 

^B1 Bar code – code 11 

 

Syntax  ^B1o,e,h,f,g 

 

Parameter 

o Orientation 

Default: N (Portrait) 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

e Number of Check digits 

Default: N for 2 digits. 

Y for 1 digit. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

Valid 

characters 

0 ~ 9 

 

Example 

 

^XA 

^FO30,70^B1,N,50,Y,Y^FD12345^FS 

^FO30,160^B1,Y,30,Y,N^FD87615^FS 

^XZ 

 

Output 

    

     
 

 



 
 

28 
 

^B2 Bar code – Interleaved 2 of 5 

 

Syntax  ^B2o,h,f,g,e 

 

Parameter 

O Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

e Check digit by mod 10 

Default: N for no check digit. 

Y for check digit. 

Valid 

characters 

0 ~ 9 

 

Example 

^XA 

^FO30,70^B2,50,Y,Y,Y^FD24680^FS 

^FO30,160^B2,30,Y,N,N^FD13579^FS 

^XZ 

 

Output 

^B3 Bar code – Code 39 

 

Syntax  ^B3o,e,h,f,g 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

e Check digit by mod 43 

Default: N for no check digit. 

Y for check digit. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

Valid 

characters 

0 ~ 9, A ~ Z, -, $, /, +, % and space. 

Example 

 

^XA 

^FO30,70^B3,,100^FDABCDZ^FS 

^XZ 

 

Output 

^B5 Bar code – Planet 

 

Syntax  ^B5o,h,f,g 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: N for no. 

Y for human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

Valid 

characters 

0 ~ 9. 

 

Example 

 

^XA 

^FO30,70^B5N,80,Y,^FD11111111111^FS 

^XZ 

 

Output 

^B7 Bar code – PDF-417 

 

Syntax  ^B7o,h,s,c,r,t 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Height for individual row 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

s Security level 

Default: 0 for level 0. 

Other: 1 to 8. 

c Number of columns to encode 

Default: 1:2 row/column aspect ratio. 

Others: 1 ~ 30. 

r Number of rows to encode 

Default: 1:2 row/column aspect ratio. 

Others: 3 ~ 90. 

t Type of guard bar at right side 

Default: N for normal. 

Other: Y for truncation. 

Example 

 

^XA 

^FO30,70^B7,5,,,,Y^FDPDF417 ABC^FS 

^FO30,150^B7,7^FDPDF417 XYZ^FS 

^XZ 

 

Output 

^B8 Bar code – EAN-8 

 

Syntax  ^B8o,h,f,g 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

Valid 

characters 

0 ~ 9.  

Exactly 7 or 8 characters are allowed. 

 

Example 

 

^XA 

^BY3^FO30,70^B8,80^FD1234567^FS 

^XZ 

 

Output 

^B9 Bar code – UPC E 

 

Syntax  ^B9o,h,f,g,e 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

e Check digit 

Default: Y for check digit. 

N for no check digit. 

Valid 

characters 

0 ~ 9.  

Exactly 10 characters are allowed. 

 

Example 

^XA 

^BY2 

^FO30,70^B9,80^FD0123456789^FS 

^FO180,70^B9,60,Y,N,N^FD0123456789^FS 

^XZ 

 

Output 

^BA Bar code – Code 93 

 

Syntax  ^BAo,h,f,g,e 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

e Check digit 

Default: Y for check digit. 

N for no check digit. 

Valid 

characters 

0 ~ 9, A ~ Z, -, $, /, +, %, &, ’, ( ) and space. 

 

Example 

^XA 

^FO30,20^BA,40^FD01234ABC89^FS 

^FO30,90^BA,30,Y,N,N^FD0123ABC789^FS 

^XZ 

 

Output 

^BC Bar code – Code 128 with subsets A, B and C 

 

Syntax  ^BCo,h,f,g,e,m 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

e Check digit 

Default: Y for check digit. 

N for no check digit. 

m Default: N for fixed mode. 

A for auto mode 

D for UCC/EAN mode 

U for UCC mode. 

Valid 

characters 

Code A: ASCII code 00H ~ 5DH. 

Code B: ASCII code 20H ~ 7FH. 

Code C: numeric only (0 ~ 9). 

Invocation 

code 

>8  FNC1 

>9  Start code A 

>:  Start code B 

>;  Start code C 

 

Example 

 

^XA 

^FO30,20^BC,50,,,,A^FD0123456789^FS 

^XZ 

 

Output 

^BD Bar code – Maxicode 

 

Syntax  ^BDm,n,t 

 

Parameter 

m Mode 

Default: 2. 

Others: 3 ~ 6. 

n Symbol number 

Default: 1. 

Others: 1 ~ 8 symbols. 

t Total number of symbols 

Default: 1. 

Others: 1 ~ 8. 
 

The ^BY command has no affect on Maxicode. The ^BD is combined with 

^FD for Maxicode. 

 

 

Syntax  ^FD<hpm><lpm> 

 

Parameter 

<hpm> High priority message (only applicable in modes 2 and 3) 

aaabbbcccccdddd for mode 2, where 

 aaa     class of service 

 bbb    country code 

 ccccc   zip code 

 dddd    zip code extension 

aaabbbcccccc for mode 3. 

<lpm> Low priority message (only applicable in modes 2 and 3) 

 

Example 

 

^XA 

^FO20,10^CVY 

^BD^FH^FD001840152382802[)>_1E01_1D961Z00004951_1

DUPSN_1D06X610_1D159_1D1234567_1D1/1_1D_1DY_1D634 

ALPHA DR_1DPITTSBURGH_1DPA_1E_04^FS 

^XZ 

 

Output 

^BE Bar code – EAN-13 

 

Syntax  ^BEo,h,f,g 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

Valid 

characters 

0 ~ 9.  

Exactly 12 or 13 characters are allowed. 

 

Example 

 

^XA 

^FO30,70^BE,80^FD123456789012^FS 

^XZ 

Output 

^BF Bar code – Micro PDF417 

 

Syntax  ^BFo,h,m 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for reverse 

landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 99 pixels. 

m Barcode mode 

Default: 0 

Barcode 

mode (m) 

Data  

Columns 

Data 

Rows 

Max 

Alphabetical  

Characters  

Max Numeric 

Characters 

0 1 11 6 8 

1 14 12 17 

2 17 18 26 

3 20 22 32 

4 24 30 44 

5 28 38 55 

6 2 8 14 20 

7 11 24 35 

8 14 36 52 

9 17 46 67 



 
 

49 
 

10 20 56 82 

11 23 64 93 

12 26 72 105 

13 3 6 10 14 

14 8 18 26 

15 10 26 38 

16 12 34 49 

17 15 46 67 

18 20 66 96 

19 26 90 132 

20 32 114 167 

21 38 138 202 

22 44 162 237 

23 4 6 22 32 

24 8 34 49 

25 10 46 67 

26 12 58 85 

27 15 76 111 

28 20 106 155 

29 26 142 208 

30 32 178 261 

31 38 214 313 

32 44 250 366 

33 4 14 20 
 

 

Example 

^XA 

^FO330,110^BY4,3,20^BFN,10,0^FDABCDEF^FS 

^FO330,310^BY4,3,20^BFN,10,0^FD12345678^FS 

^XZ 

 

Output 

^BI Bar code – Industrial 2 of 5 

 

Syntax  ^BIo,h,f,g 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

Valid 

characters 

0 ~ 9.  

 

Example 

 

^XA^BY3 

^FO30,70^BI,50^FD24680^FS 

^XZ 

Output 

^BJ Bar code – Standard 2 of 5 

 

Syntax  ^BJo,h,f,g 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

Valid 

characters 

0 ~ 9.  

 

Example 

 

^XA^BY3 

^FO30,70^BJ,50^FD24680^FS 

^XZ 
 

Output 

^BK Bar code – Codabar 

 

Syntax  ^BKo,e,h,f,g,k,l 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

e Check digit 

Default: N 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

k Start code 

Default: A. 

Others: B,C,D,*,N,E or T 

l Stop code 

Default: A. 

Others: B,C,D,*,N,E or T 

Valid 0 ~ 9, A, B, C, D, E, T, N, *, -, :, ., $, /, + 

characters 
 

Example 

 

^XA 

^FO40,70^BK,N,50,,,A,B^FD2480^FS 

^XZ 

 

Output 

^BL Bar code – Logmars 

 

Syntax  ^BLo,h,g 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

Valid 

characters 

0 ~ 9, A ~ Z, -, ., *, $, /, + , , and space 

 

Example 

 

^XA 

^FO40,70^BL,50^FDABC2^FS 

^XZ 

 

Output 

^BM Bar code – MSI 

 

Syntax  ^BMo,e,h,f,g,h 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

e Check digit 

Default: B (1 mod 10) 

Others: A (no check digit), C (2 mod 10), D (1 mod 10 and 

1 mod 11). 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

h Print check digit 

Default: N for no. 

Other: Y for yes. 

Valid 

characters 

0 ~ 9 

 

Example 

 

^XA 

^FO40,70^BM,,50,,,Y^FD9876^FS 

^XZ 

 

Output 

^BO Bar code – Aztec 

 

Syntax  ^BOa,b,c,d,e 

 

Parameter 

a Orientation 

Default: N (Portrait) 

b Magnification factor 

Default: 2 for 200 dpi printers; 3 for 300 dpi printers. 

Others: 1~10. 

c ECIC (Extended channel interpretation code) indicator 

Default: N (Data includes ECIC) 

Y (Data does not include ECIC) 

d Error correction level & Symbol size/barcode type indicator 

Default: 0 (error correction level) 

Others: 01~99: error correction percentage 

101~104: 1~4-layer compact symbol 

201~232: 1~32-layer full-range symbol 

300: a simple Aztec “Rune” 

e Menu symbol indicator 

Default: N is for it to be a menu symbol. 

Y is not for it to be a menu symbol. 

 

Example 

^XA 

^FO20,250^BON,6,N,0,N^FDAztec barcode test^FS  

^XZ 

 

Output 

^BP Bar code – Plessey 

 

Syntax  ^BPo,e,h,f,g 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

e Print Check digit. 

Default: N for No. 

Others: Y for Yes. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: Y for text above barcode. 

N for text below barcode. 

Valid 

characters 

0 ~ 9, A~F. 

Maximum: 100 characters. 

 

Example 
 

^XA 

^FO50,100^BPN,N,80,Y,N^FD12345^FS 

^XZ 

 

Output 

^BQ Bar code – QR code 

 

Syntax  ^BQo,e,f^FDnm,c<data string> 

 

Parameter 

o Orientation 

Default: N. No rotation is available. 

e Model 

Default: 2 for enhanced. 

Others: 1 for original. 

f Magnification factor 

Default: 2 

Others: 1~10. 

n Error correction level 

Default: M for standard level. 

H for ultra-high reliability level. 

Q for high reliability level. 

L for high density level. 

m Data input modes 

Default: A for automatic 

M for manual input. 

c Character mode 

N for numeric. 

A for alphanumeric. 

Bxxxx for 8-bit byte mode; xxxx is indicate number of 

characters. (A Simple or Traditional Chinese word is 2 

characters) 

K for Kanji characters. 

^BR RSS Bar Code (GS1 DataBar) 

 

Syntax  ^BRa,b,c,d,e,f 

 

Parameter 

a Orientation 

Default: R for reverse landscape 

Others: N (Portrait), I for reverse portrait and B for landscape. 

b Bar code types 

Default: 1 

Others:  

1= RSS14 (GS1 DataBar Omnidirectional) 

2= RSS14 Truncated (GS1 DataBar Truncated) 

3= RSS14 Stacked (GS1 DataBar Stacked) 

4= RSS14 Stacked Omnidirectional (GS1 DataBar Stacked 

Omnidirectional) 

5= RSS Limited (GS1 DataBar Limited) 

6= RSS Expanded (GS1 DataBar Expanded) 

7= UPC-A Composite 

8= UPC-E Composite 

9= EAN-13 Composite 

10= EAN-8 Composite 

11= UCC/EAN-128 Composite (CC-A/B) 

12= UCC/EAN-128 Composite (CC-C) 

c Magnification factor 

Default: 1 

Others: 1 to 10 



 
 

68 
 

d Separator height 

Always value: 1 

e Bar code height. (Only affect the linear portion of the bar 

code.) This parameter will not be effective by the parameter 

“c.” 

Default value: 

RSS14=33 

RSS14 Truncated=13 

RSS14 Stacked=13 

RSS14 Stacked Omnidirectional=69 

RSS Limited=10 

RSS Expanded=34 

Others:1 to 32000 dots 

f The segment width (RSS expanded only) 

Default: 22 

Others:2 to 22 (even numbers only in segments per line) 

 

Example 1 

 

^XA 

^MNN 

^FO10,10^BRN,1,3,1,33,22^FD1234567890|RSS14^FS 

^FO10,140^AC^FDRSS14^FS 

 

^FO10,220^BRN,2,3,1,13,22^FD89121121225|RSS14 

Truncated^FS 

^FO10,310^AC^FDRSS14 Truncated^FS 

^FO410,10^BRN,3,3,1,13,22^FD88689121124|RSS14 

Stacked^FS 

^FO410,170^AC^FDRSS14 Stacked^FS 

 

^FO10,400^BRN,4,2,1,69,22^FD0968851800|RSS14 

Stacked Omnidirectional^FS 

^FO10,650^AC^FDRSS14 Stacked Omnidirectional^FS 

 

^FO410,250^BRN,5,3,1,10,22^FD911006072123|RSS14 

Limited^FS 

^FO410,340^AC^FDRSS14 Limited^FS 

 

 

^FO410,410^BRN,6,2,1,34,22^FD8860227662000|RSS14 

Expanded^FS 

^FO410,510^AC^FDRSS14 Expanded^FS 

^XZ 

 

Output 1 

 



 
 

70 
 

 

 

Example 2 

 

^XA 

^FO100,100,0^BY3^BRN,7,2,2,25^FD12345678901|UPC-A 

Composite^FS 

^FO110,300^ADN,18,10^FDUPC-A Composite^FS 

 

^FO100,400,0^BY3^BRN,8,2,2,25^FD1230000045|UPC-E 

Composite^FS 

^FO110,620^ADN,18,10^FDUPC-E Composite^FS 

^FO100,700,0^BY3^BRN,9,2,2,25^FD1234567|EAN-13 

Composite^FS 

^FO110,900^ADN,18,10^FDEAN-13 Composite^FS 

 

^FO400,100,0^BY3^BRN,10,2,2,25^FD123456789012|EAN

-8 Composite^FS 

^FO400,280^ADN,18,10^FDEAN-8 Composite^FS 

 

^FO400,400,0^BY3^BRN,11,3,2,20^FD1234567|UCC/EAN-

128 Composite CC-A/B^FS 

^FO400,540^ADN,18,10^FDUCC/EAN-128 Composite 

CC-A/B^FS 

 

^FO400,700,0^BY3^BRN,12,2,2,25^FD1234567|UCC/EAN-

128 Composite CC-C^FS 

^FO400,850^ADN,18,10^FDUCC/EAN-128 Composite 

CC-C^FS 

^PQ1,0,1,Y 

^XZ 

 

Output 2 

^BS Bar code – UPC/EAN extension 

 

Syntax  ^BSo,h,f,g 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: Y for text above barcode. 

N for text below barcode. 

Valid 

characters 

0 ~ 9. 

Exactly 2 or 5 characters. 

 

Example 

 

^XA^FO30,70^BS,50^FD98^FS 

^FO160,70^BS,50,Y,N^FD98765^FS 

^XZ 

Output 

^BT Bar code – TLC39 

 

Syntax  ^BTo,w1,r1,h1,w2,h2 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

w1 Bar code width ratio of Code 39 

Default: 2  

Others: 1 ~10 pixels 

r1 Bar code height of Code 39 

Default: 2.0  

Others: 2.0 ~3.0 pixels 

h1 Bar code width of Code 39 

Default: 40 (200 dpi printer); 60 (300dpi printer)  

Others: 1 ~ 999 pixels 

w2 Row width of Micro PDF417 

Default: 2 

Others: 1 ~10 pixels 

h2 Row height of Micro PDF417 

Default: 4 

Others: 1 ~255 pixels 

Valid 

characters 

TLC39 contains ECI NO. , Serial NO. and Additional data. 

ECI Number:  

Code 39 barcode and it must be 6 digits. 

If the 7th digit is not a comma, only prints Code 39(only the 

first 6 digits print).  

Serial Number: 

The Serial number is printed as Micro PDF417. 

The Serial number contains up to 25 digits. 

A comma follows the Serial number. 

Additional data: 

Additional data is used as a country code. 

The data contains up to 150 bytes, which is included serial 

number comma. 

 

Example 

 

^XA^FO30,70^BY3,2.3^BTN,2,2,100,2,4 

^FD123456,ABCDEFGHIJ123456789012345,012^FS 

^XZ 

 

Output 

^BU Bar code – UPC A 

 

Syntax  ^BUo,h,f,g,e 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: Y for yes. 

N for no human readable text. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

e Print check digit 

Default: Y for yes. 

Other: N for no. 

Valid 

characters 

0 ~ 9. 

Exactly 11 or 12 characters. 

 

Example 

 

^XA 

^FO40,70^BU,50Y^FD01234567890^FS 

^XZ 

 

Output 

^BX Bar code – Data Matrix 

 

Syntax  ^BXo,h,s,c,r,f,g 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Dimensions 

Default: set by ^BY command. 

s Quality Level 

Accept values: 0, 50, 80, 100, 140 and 200. 

c Columns to encode. (Auto) 

r Rows to encode. (Auto) 

f Format ID 

g Escape sequence control character. 

Default: “ _ ”. 

Other: Any characters. 
 

Example 

 

^XA^FO100,100^BXN,10,200,,,, 

^FDARGOX INFORMATION^FS 

^XZ 

 

Output 

^BY Parameter defaults for general bar codes 

 

Syntax  ^BYw,r,h 

 

Parameter 

w Width of narrow bar 

Default: 2 pixels (cold start). 

Others: 1 ~ 10 pixels. 

r Bar ratio 

Default: 3.0 (cold start). 

Others: 2.0 ~ 3.0. 

h Barcode height 

Default: 10 pixels. 

Others: 1 ~ 999 pixels. 

 

Example 

 

^XA^BY2,3.0,40^FO30,40^BI^FD24680^FS 

^BY3,2.5,60^FO30,130^BI^FD24680^FS 

^XZ 

 

Output 

^BZ Bar code – Postal 

 

Syntax  ^BZo,h,f,g,t 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

h Bar code height 

Default: set by ^BY command. 

Others: 1 ~ 999 pixels. 

f Human readable text 

Default: N for no human readable text. 

Y for yes. 

g Position for human readable text 

Default: N for text below barcode. 

Y for text above barcode. 

t Barcode type 

Default: 0 for Postnet bar code. 

1 for Planet bar code. 

2 for Reserved. 

3 for USPS Intelligent Mail bar code. 

 

Valid 

characters 

0 ~ 9. 

Example 

 

^XA^FO40,70^BZN,50,Y,N,0^FD12345^FS 

^FO40,30^AE^FDPostnet bar code^FS 

^FO40,240^BZN,50,Y,N,1^FD12345678901^FS 

^FO40,200^AE^FDPlanet bar code^FS 

^FO40,410^BZN,50,Y,N,3^FD00123123456123456789^FS 

^FO40,365^AE^FDUSPS Intelligent Mail bar code^FS ̂ XZ 

 

Output 

^CC Changes the caret character 

 

Syntax  ^CCx  or  ~CCx 

 

Parameter 

x New caret character 

The default caret ^ will be replaced by x where x is a printable 

character. 

 

Description This new caret will be saved to the flash or E2PROM 

permanently. 

^CD Changes the delimiter character 

 

Syntax  ^CDx  or  ~CDx 

 

Parameter 

x New delimiter character 

The default delimiter will be replaced by x where x is a 

printable character. 

 

Description This new delimiter will be saved to the flash or E2PROM 

permanently. 

 

Example 

 

~CD* 

This command changes the delimiter from ‘,’ to ‘*’. 

^CF Changes default font 

 

Syntax  ^CFc,h,w 

 

Parameter 

c Default font ID 

Default: A. 

Others: B ~ H, 0 ~ 9. 

h Character cell height in pixels. 

0 ~ 999. 

w Character cell width in pixels. 

0 ~ 999. 

 

Description If the specified font does not exist, the default font will be 

selected. This default font ID will be saved to the flash or 

E2PROM permanently. 

 

Example 

 

^XA^CFF^XZ 

^XA 

^FO30,100^AZ^FDFONT F ?^FS 

^XZ 

 

Output 

^CI Changes symbol set 

 

Syntax  ^CIc 

 

Parameter 

c Symbol set 

Default: 0 for USA. 

Others: 1 ~ 13. 

17 for Deprecated - UCS-2 Big Endian; only with ^F8 (UTF-8 encoding). 

26 for Multibyte Asian Encodings with ASCII Transparency. 

   - The encoding is controlled by the conversion table (*.DAT). The 

correct table must be present for the conversion to function. 

   - CI26 supports ASCII transparency for Asian encodings. 7F and less are 

treated as single byte characters. 80 to FE is treated as the first byte 

of a 2 byte character 8000 to FEFF in the encoding table for Unicode. 

27 for Code page 1252 

28 for UTF-8 encoding - Unicode Character Set. 

29 for UTF-16 Big-Endian encoding - Unicode Character Set. 

    - This set is effective when the encoding is between ^FD and ^FS. 

30 for UTF-16 Little-Endian encoding - Unicode Character Set. 

- This set is effective when the encoding is between ^FD and ^FS. 

31 for Code page 1250 

33 for Code page 1251 

34 for Code page 1253 

35 for Code page 1254 

36 for Code page 1255 

This default font will be saved to the flash or E2PROM permanently. 

 

 23H 30H 40H 5BH 5CH 5DH 5EH 60H 7BH 7CH 7DH 7EH Country 

CI0 #  @ [ ¢ ] ^ ʽ { ¦ } ~ USA 

CI1 # 0 @ ⅓ ¢ ⅔ ^ ʽ ¼  ½  ¾  ~ USA2 

CI2 £  @ [ ¢ ] ^ ʽ { ¦ } ~ UK 

CI3 ƒ 0 § [ Ĳ ] ^ ʽ { ĳ } ~ Holland 

CI4 # 0 @ Æ  Ø  Å ^ ʽ æ  ø å ~ Den/Nor 

CI5 Ü  0 É Ä Ö  Å Ü  é ä ö å ü Swe/Fin 

CI6 # 0 § Ä Ö  Ü  ^ ʽ ä ö ü β German 

CI7 £  à [ ç ] ^ ʽ é ¦ ù è France1 

CI8 # 0 à â ç ê î ô é ù è û France2 

CI9 £  § [ ç é ^ ù à ò è ì Italy 

CI10 # 0 § ¡ Ñ  ¿ ^ ʽ { ñ ç ~ Spain 

CI11 £  É Ä Ö  Ü  ^ ä ë ï ö ü Misc. 

CI12 #  @ [ ¥ ] ^ ʽ { ¦ } ~ Japan 

CI13 # 0 @ [ \ ] ^ ʽ { ¦ } ~ Page850 

^CT Changes the tilde character 

 

Syntax  ^CTx  or  ~CTx 

 

Parameter 

x New tilde character 

The default tilde ’ ~‘ will be replaced by x where x is a 

printable character. 

 

Description This new tilde will be saved to the flash E2PROM 

permanently. 

 

Example 

 

~CT= 

 

This command changes the tilde from ‘~’ to ‘=’. 

^CW Defines the font ID 

 

Syntax  ^CWc,d:f 

 

Parameter 

c New font ID 

Range: A ~ Z, 0 ~ 9. 

By the new font ID, the font can be accessed. 

d Device driver 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

f File name 

8 character file name with 3 characters extension name. 

 

Example 

 

^XA^CWZ,COMIC.FNT 

^FO30,100^AZ,30,25^FDFont Z for Comic^FS 

^XZ 

 

Output 

~DB Downloads a bitmap font 

 

Syntax  ~DBd:f,o,h,w,base,hmi,#char,msg,data 

 

Parameter 

d Destination device driver 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

f File name 

8 character file name with 3 character extension name. 

The default extension name for font is “.FNT”. 

o Orientation of current font 

Always N for portrait. 

h Cell height in pixels 

w Cell width in pixels 

base Base line position in pixels from top 

hmi Horizontal motion index 

The movement distance for a space or non-printable code. 

#char Total number of character to be downloaded 

msg A maximum of 63 characters for general message. 

data Parameters and image for characters 

For a character its parameters and image are defined as follow. 

 

#code.h.w.xoff.yoff.deltax.image 

 

code  Character code. 

h  Character height in pixels. 

w  Character width in pixels. 

xoff  Left offset in pixels. 

yoff  Top offset in pixels. 

deltax Cursor movement in pixels after the character is printed. 

image All image is represented by the hex code. 

~DE Downloads encoding 

 

Syntax  ~DEd:f,t,data 

 

Parameter 

d Device driver 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

f Table name 

8 character file name with 3 characters extension name. 

The default extension name for table is “.DAT”. 

t Total number of bytes in table required. 

data Table date 

A string of ASCII hexadecimal values. 

 

Example 

 

~DER:TABLE.DAT,29998,300021213001... 

 

(29998 two-digit hexadecimal values) 

^DF Downloads form 

 

Syntax  ^DFd:f 

 

Parameter 

d Device driver 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

f File name 

8 character file name with 3 characters extension name. 

The default extension name is “.PPLZ”. 

 

Description This command ^DF is used to define the form including 

variables, counters. The form is executed only by ^XF 

command. Refer to the ^XF command and its example for 

details. 

~DG Downloads graphic 

 

Syntax  ~DGd:f,t,w,data 

 

Parameter 

d Device driver 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

f File name 

8 character file name with 3 characters extension name. 

The default extension name for graphic is “.GRF”. 

t Total number of bytes in graphic 

w 156 bytes per raster line 

data Rater data 

The data are represented by hex codes and control codes for 

compression. 

 

t (total no. of raster data) = 

w (width of raster line) * h (no. of raster lines) 

 

Example 

 

~DGTRIANGLX,42,6, 

F00000000000 

FF0000000000 

FFF000000000 

FFFF00000000 

FFFFF0000000 

FFFFFF000000 

FFFFFFFFFFFF 

^XA 

^FO30,40^XGTRIANGLX,1,1^FS 

^FO40,60^XGTRIANGLX,5,5^FS 

^XZ 

 

Output 

~DN Abort Download Graphic 

 

Syntax  ~DN 

 

Dscription This command is used to clear download graphic in the 

printer. 

~DU Downloads unbounded TrueType font 

 

Syntax  ~DUd:f,s,data 

 

Parameter 

d Device driver 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

f File name 

8 character file name with 3 characters extension name. 

The default extension name for TrueType is “.FNT”. 

s Font size 

data TrueType data 

~EF Erases all stored forms 

 

Syntax  ~EF 

 

Description After the command is received all forms stored in the 

printer will be erased. 

~EG Erases all stored graphics 

 

Syntax  ~EG 

 

Description After the command is received all graphics stored in the 

printer will be erased. 

^FA Field allocate 

 

Syntax  ^FAm 

 

Parameter 

m Character Memories to be saved 

Default: 1. 

Others: 1 ~256. 

 

Description This command is used to let the printer allocate memory 

spaces for the field to be saved. 

^FB Defines the format of a block data 

 

Syntax  ^FBa,b,c,d,e 

 

Parameter 

a Width of field in pixels 

Default: 0. 

Others: 1 ~ 999. 

b Maximum number of lines 

Default: 1. 

Others: 2 ~ 999. 

c Extra line space in pixels between lines 

Default: 0. 

Others: -999 ~ 999. 

d Justification of text 

Default: L. 

Others: C (center), J (margin to margin) and R (right). 

e Secondary left margin 

Default: 0. 

Others: 1 ~ 999. 

 

Description Each block starts from ^FB and ends by an ^FS. Some 

special control codes in the ^FD command are used for 

special functions. 

 

\& carriage return and line feed 

\(*) soft hyphen (word break with a dash) 

\\ Same as \ 

Example 

 

^XA 

^AF^FO20,20^FB220,6 

^FDThis is a test for FB command^FS 

^XZ 

 

Output 

^FC Field Clock 

 

Syntax  ^FCa,b,c 

 

Parameter 

a Primary clock indicator character 

Default: % 

Others: ASCII character 

b Second clock indicator character 

Default: None(The value can’t be the same as a or c) 

Others: ASCII character 

c Third clock indicator character 

Default: None(The value can’t be the same as a or b) 

Others: ASCII character 

 

Description The command is for setting the clock-indicators and the 

clock mode. Besides, this command is also included within 

each label field command string each time the Real-Time 

Clock values are required within the field.  

 

Example 

 

^XA 

^FO20,100^A0N,50,50 

^FC!,#,% 

^FDPrimary: !Y/!b/!d/!H/!M^FS 

^FO20,200^A0N,50,50 

^FC!,#,% 

^FDSecondary: #Y/#b/#d/#H/#M^FS 

^FO20,300^A0N,50,50 

^FC!,#,% 

^FDThird: %Y/%b/%d/%H/%M^FS 

^XZ 

 

Output 

Command Characters 

Command 

Character 

Function Value 

%a The abbreviated weekday 

name 

 

%A The weekday name  

%b The abbreviated month name  

%B The month name  

%d The day of the month number  01~31 

%H The hour of the day-military  00~23 

%I The hour of the day-civilian 01~12 

%j The day of the year 001~366 

%m The month number  02~12 

%M The minute  00~59 

%p AM or PM designation  

%S The seconds  00~59 

%U The week of the year  00~53, Sunday is the first day 

%W The week of the year  00~53, Monday is the first day 

^w The day of the week  0(Sunday)~6(Saturday) 

%y The 2 digits of the year  

%Y The 4 digits of the year  

^FD Field data 

 

Syntax  ^FDdata 

 

Parameter Data for text or barcode 

 

Description In general this command is used to include the data for 

text or barcode and ends by a ^FS command. Some special 

control codes in the ^FD command are used for special 

functions. 

 

\& carriage return and line feed 

\(*) soft hyphen (word break with a dash) 

\\ Same as \ 

^FH Field HEX 

 

Syntax  ^FHa 

 

Parameter 

a Character before a hexadecimal code 

Default: _. 

Others: printable characters. 

 

Description For some control codes or un-printable codes this can 

make you print them. This command must be present 

before the ^FD command. 

 

Example 

 

^XA 

^FO40,60^AE^FH^FD_7e for 7EH^FS 

^FO40,110^AE^FH\^FDA \7e for 7EH^FS 

^XZ 

 

Output 

^FN Field number 

 

Syntax  ^FNa 

 

Parameter 

a ID Number to be assigned 

Default: 0. 

Others: 1 ~ 999. 

 

Description The ^FN is usually used in a form. By this you may specify 

the field number and fill it with updated data. 

 

Example 

 

^XA^DFFMT^FS 

^BY2,3,100 

^FO50,60^AF^FN1^FA9^FS 

^FO50,110^B3^FN2^FA6^FS 

^XZ 

 

^XA^XFFMT^FS 

^FN1^FDNew Data^FS 

^FN2^FD123^FS 

^XZ 

 

Output 

^FO Moves the position for print pattern 

 

Syntax  ^FOx,y,z 

 

Parameter 

x Horizontal coordinate 

Default: 0. 

Others: 1 ~ 65535 printing width of the printer. 

y Vertical coordinate 

Default: 0. 

Others: 1 ~ 65535 printing length of the printer. 

z Justification 

Default: 0= left justification 

Others:1= right justification; 2= justification (script 

dependent) 
 

Description The location of ^FO is relative to the origin set by the ^LH 

command. (Please also refer to related commands- ^FP 

and ^FT.) 

 

Example1 

 

XA 

^FO50,50^AB^FD(50,50)^FS 

^FO150,150^AB^FD(150,150)^FS 

^XZ 

 

Output 1 

Example 2 

 

^XA 

^LL560 

^PW800 

~SD16 

^MD0 

^PR3,3,3 

^FO200,300^GB400,1,2,B^FS 

^FO400,150^GB1,300,2,B^FS 

^FO400,300,0^FPH^AVN,,^FDvertiNN^FS 

^XZ 

 

Example 3 

 

^XA 

^LL560 

^PW800 

~SD16 

^MD0 

^PR3,3,3 

^FO200,300^GB400,1,2,B^FS 

^FO400,150^GB1,300,2,B^FS 

^FO400,300,1^FPH^AVN,,^FDvertiNN^FS^XZ 

 

Output 2 & Output 3 

 

Output 2  

(Left justified) 

Output 3 

(Right justified) 

^FP Field Parameter 

 

Syntax  ^FPd,g 

 

Parameter 

d Direction 

Default: H for horizontal printing (left to right)       

Others: 

R for vertical printing (top to bottom) 

V for vertical printing (right to left) 

g Extra inter-character gap in pixels 

Default: 0. 

Others: 0 ~ 9999 

 

Description This command enables vertical and reverse formatting of 

the font field. 

 

Example 

 

^XA 

^LL560 

^PW800 

~SD16 

^MD0 

^PR3,3,3 

^FO200,300^GB400,1,2,B^FS 

^FO400,150^GB1,300,2,B^FS 

^FO400,300,0^FPH^AVN,,^FDFPH^FS 

^XZ 

 

^XA 

^LL560 

^PW800 

~SD16 

^MD0 

^PR3,3,3 

^FO200,300^GB400,1,2,B^FS 

^FO400,150^GB1,300,2,B^FS 

^FO400,300,0^FPR^AVN,,^FDFPR^FS 

^XZ 

 

^XA 

^LL560 

^PW800 

~SD16 

^MD0 

^PR3,3,3 

^FO200,300^GB400,1,2,B^FS 

^FO400,150^GB1,300,2,B^FS 

^FO400,300,0^FPV^AVN,,^FDFPV^FS 

^XZ 

Output 

^FR Reverse print 

 

Syntax  ^FR 

 

Description This command enables reverse print for the next field 

data. 

 

Example 

 

^XA 

^FO40,20^GB80,0,100^FS 

^FO40,55^FR^A0,40,35^FDReverse^FS 

^XZ 

 

Output 

^FS End of field 

 

Syntax  ^FS 

 

Description This command denotes the end of field. It can also be 

represented by a single control code (0FH). 

^FT Defines coordinate for print pattern 

 

Syntax  ^FTx,y,z 

 

Parameter 

x Horizontal coordinate 

Default: 0. 

Others: 1 ~ 65535. 

y Vertical coordinate 

Default: 0. 

Others: 1 ~ 65535. 

z Justification 

Default: 0 

Others: 0= left justification; 1= right justification; 2= auto 

justification 

 

Description Similar to ^FO command the location of ^FO is relative to 

the origin set by the ^LH command but the object position 

is different. 

 

Example 1 

 

^XA 

^FT100,100^A0,50,40^FDFT^FS 

^FO100,100^GB100,0,2^FS 

^FO100,100^A0,50,40^FDFO^FS 

^XZ 

Output 1 

 

     

 

Example 2 

 

^XA 

^LL560 

^PW800 

~SD16 

^MD0 

^PR3,3,3 

^FO200,300^GB400,1,2,B^FS 

^FO400,150^GB1,300,2,B^FS 

^FT400,300,0^FPH^AVN,,^FDvertiNN^FS 

^XZ 

 

Example 3 

 

^XA 

^LL560 

^PW800 

~SD16 

^MD0 

^PR3,3,3 

^FO200,300^GB400,1,2,B^FS 

^FO400,150^GB1,300,2,B^FS 

^FT400,300,1^FPH^AVN,,^FDvertiNN^FS 

^XZ 

 

Output 2 & Output 3 

 

Output 2  

(Left justified) 

Output 3 

(Right justified) 

^FV Defines variable 

 

Syntax  ^FVdata 

 

Parameter Data to be printed 

 

Description The data in ^FV command can be replaced at next label. In 

general this command is combined with ^MCN command 

since there are still other fixed data that cannot be cleared 

until the last label. 

 

Example 

 

^XA 

^FT50,50^AB^FDFIXED DATA^FS 

^FT50,80^A0,30,20^FVVARIABLE^FS 

^MCN 

^XZ 

 

^XA 

^FT50,80^A0,30,20^FVNEW DATA^FS 

^MCY 

^XZ 

 

Output 

^FW Defines default orientation 

 

Syntax  ^FWo 

 

Parameter 

o Orientation 

Default: N (Portrait). 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

 

Description The ^FW command defines the default orientation for all 

fields. If a command has a specific orientation parameter 

the default one is not used. 

 

Example 

 

^XA 

^FWR 

^FO40,50^AB^FDTEXT 1^FS 

^FO100,100^AC^FDTEXT 2^FS 

^XZ 

 

Output 

^FX Comment 

 

Syntax  ^FWdata 

 

Parameter Data for comment 

 

Description The ^FX command is just for human read only and ignored 

by the printer. It allows user to add some “non-printing” 

information within a label format. 

 

Example 

 

^XA 

^FO40,50^AB^FDNON-COMMENT1^FS^FX!COMMENT 

^FO40,100^AC^FDNON-COMMENT2^FS 

^XZ 

 

Output 

^GB Box or line 

 

Syntax  ^GBw,h,t,c 

 

Parameter 

w Width of box or line in pixels 

Default: 1. 

Others: 0 ~ 999. 

h Height of box or line in pixels 

Default: 1. 

Others: 0 ~ 999. 

t Thickness in pixels 

Default: 1. 

Others: 0 ~ 999. 

c Color 

Default: B. 

Other: W for white or clear. 

 

Example 

 

^XA^FO50,50^GB100,90,3^FS 

^FO60,70^GB50,0,10^FS 

^XZ 

 

Output   

^GD Diagonal line 

 

Syntax  ^GDw,h,t,c,o 

 

Parameter 

w Width of diagonal line in pixels 

Default: 3 or t (thickness). 

Others: 3 ~32000. 

h Height of diagonal line in pixels 

Default: 3 or t (thickness). 

Others: 3 ~ 32000. 

t Thickness in pixels 

Default: 1. 

Others: 1 ~ 32000. 

c Color 

Default: B. 

Other: W for white or clear. 

o Orientation 

Default: R  

Others: R (or /) for right-leaning, L (or \) for left-leaning. 

 

Example 

 

^XA^FO20,20^FO20,20^GD330,183,1,,R^FS 

^FO20,20^GD330,183,1,,L^FS 

^FO20,20^GB332,184,1^FS 

^XZ 

Output 

^GF Direct graphic 

 

Syntax  ^GFf,b,t,w,data 

 

Parameter 

f Format 

Default: A for ASCII (Hex). 

Others: B for binary. 

b Binary byte count 

In general b is same as t. 

t Total byte count 

This depends on the area of graphic. 

w Byte count of a raster line 

data Raster data 

Hex or binary data according to the specified format. 

 

Description Such graphic is loaded directly into the image frame buffer. 

No graphic name is specified. 

 

Example 

 

^XA 

^FO50,60^GFA,100,100,5 

FFFFFFFFFFFFFFFFFFFF 

FFFF00FFFFFFFF00FFFF 

FFF0000FFFFFF0000FFF 

FF000000FFFF000000FF 

F00000000FF00000000F 

FF000000FFFF000000FF 

FFF0000FFFFFF0000FFF 

FFFF00FFFFFFFF00FFFF 

FFFFFFFFFF000FFFF000 

00FFFFFF000FFFFFFFF0 

^XZ 

 

Output 

^GS Prints special symbols 

 

Syntax  ^GSo,h,w 

 

Parameter 

O Orientation 

Default: N (Portrait) 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

H Character height in pixels 

Same definition as that in ^A. 

W Character width in pixels 

Same definition as that in ^A. 

 

Description There are only five special symbols denoted by A ~ E. 
 

Example 

 

^XA^FO50,50^GS^FDABCDE^FS 

^XZ 

 

Output 

^HF Host Format 

 

Syntax  ^HFd,o,x 

 

Parameter 

d Device to recall file 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

o File name 

Alphanumeric characters 

x .PPLZ 

 

Description This command sends stored formats to the host.  

 

Example 

 

Download this code to the printer through a terminal emulator. 

 

^XA 

^DFB:ARGOX.PPLZ 

^FO50,50^A100,100 

^FDHF TEST^FS 

^XZ 

 

Then send this code to the printer. 

 

^XA 

^HFB:ARGOX.PPLZ 

^XZ 

 

The terminal emulator will return this code. 

 

^XA^DFB:ARGOX.PPLZ 

^FO50,50^A100,100^FDHF TEST^FS 

^XZ 

^HH Configuration Label Return 

 

Syntax  ^HH 

 

Description After sending this command, the configuration label will 

return to PC through the com port.  

 

Example 

 

^XA^HH^XZ 

~HI Host identification 

 

Syntax  ~HI 

 

Description Send this command through the RS232 the printer will 

response with data for identification. 

 

Example 

~HI 

 

Response 

 

B-FV4,V1.0.0,8,32768KB 

 

where 

B-FV4  model name. 

V1.0.0  firmware version. 

8   resolution, 8 pixels per millimeter. 

32768KB standard RAM size. 

~HM Memory status 

 

Syntax  ~HM 

 

Description Send this command through the RS232 the printer will 

response with data for the status of memory. 

 

Example 

 

~HM 

 

Response 

 

32768,3682,3682 

 

where 

32768  total size in KB of memory. 

3682  Maximum free size of memory. 

3682  Currently available memory size. 

~HS Host status 

 

Syntax  ~HS 

 

Description Send this command through the RS232, the printer will 

response with data for the configuration and settings of 

the printer. 

 

Example 

 

~HS 

 

Response 

 

014,0,0,0251,000,0,0,0,000,0,0,0  

000,0,0,0,0,2,5,0,00000000,1,000  

0000,0  

 

stands for code 02H<STX> and  for code 03H<ETX>. 

 

 

There are three string groups in the response. 

 

String 1 

<STX>aaa,b,c,dddd,eee,f,g,h,iii,j,k,l<ETX><CR><LF> 

String 2 

<STX>mmm,n,o,p,q,r,s,t,uuuuuuuu,v,www<ETX><CR><LF> 

String 3 

<STX>xxxx,y<ETX><CR><LF> 

 

 

aaa   RS232 settings 

  This value is a 3 digit decimal representation of a 0 ~ 8 number. 

  aaa = a8a7a6a5a4a3a2a1a0 

a8 baud rate. 0 for less than or equal to 19200, 1 

for 38400. 

   a7  flow control. 0 for Xon/Xoff and 1 for DTR. 

   a6  Parity. 0 for odd parity and 1 for even parity. 

a5 Parity control. 0 for no parity and 1 for parity 

enabled. 

   a4  Stop bit number. 0 for 2 bits and 1 for 1 bit. 

a3 Data bit number. 0 for 7 bits and 1 for 8 data 

bits. 

a2a1a0 Baud rate. 010-600, 011-1200, 100-2400, 

101-4800,110-9600, 111-19200. Baud rate is 

38400 when a8=1 and a2a1a0=001. 

b  Paper out flag. 0 for normal. 

c  Pause flag. 1 for pause mode. 

dddd Label length. 

eee  Number of forms in the printer. 

f  Buffer full flag. 

g  Communication diagnostic mode. Always 0. 

h  Reserved. Always 0. 

iii  Reserved. Always 000. 

j  Reserved. Always 0. 

k  Reserved. Always 0. 

l  Reserved. Always 0. 

 

mmm Modes 

This value is a 3 digit decimal representation of a 0 ~ 7 number. 

  mmm = m7m6m5m4m 3m 2m1m0 

   m7 Media type. 0 for die-cut and 1 for continuous. 

   m6 reserved. Always 0. 

   m5 reserved. Always 0. 

   m4m3m2m1 reserved. Always 000. 

   m0 Ribbon mode. 0 for DT and 1 for TT. 

n   Reserved. Always 0. 

o   Reserved. Always 0. 

p   Ribbon flag. 1 for ribbon out. 

q   Thermal transfer mode. 1 for TT (thermal transfer). 

r   Print mode. 

0 rewind 

1 peel off 

2 tear off 

s   Print width. 

t   Label waiting flag. 1 for waiting in peel-off mode. 

uuuuuuuu Number of labels left for printing. 

v   reserved. Always 1 

www  Number of graphic objects in the printer. 

Xxxx   Reserved. Always 0. 

y   Reserved. Always 0. 

^HW Directory list 

 

Syntax  ^HWd:o.e 

 

Parameter 

d Device driver name 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

o Object name 

Default: *. 

e Extension name 

Default: *. 

 

Description The function is similar to the DIR command under MS-DOS. 

It lists the files through the RS232 that meet the file 

specification. 

 

Example 

 

^XA^HW^XZ 

 

Response 

 

 

- DIR R:*.* 

* R:Main.PPLZ      008192      

- 01404544 bytes free R RAM: 

stands for code 02H<STX> and  for code 03H<ETX>. 

^ID Erases graphics 

 

Syntax  ^IDd:o.e 

 

Parameter 

d Device driver name 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

o Object name 

Default: *. 

e Extension name 

Default: “.GRF”. 

 

Example 

 

^XA^IDR:*.*^XZ 

 

This example deletes all object files. 

^IL Load stored images 

 

Syntax  ^ILd:o.e 

 

Parameter 

d Device driver name 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

o Object name 

Default: *. 

e Extension name 

Default: “.GRF”. 

 

Description The function is used to load a stored image of a form and 

merge it with additional data. 

 

Example 

  

^XA^CFD 

^FO50,20^AB^FDORIGINAL DATA0^FS 

^FO50,60^AB^FDORIGINAL DATA1^FS 

^ISR:SAM.GRF,N 

^XZ 

 

^XA^ILR:SAM.GRF^FS 

^FO50,100^AB^FDNEW DATA^FS 

^XZ 

Output 

^IM Moves image 

 

Syntax  ^IMd:o.e 

 

Parameter 

d Device driver name 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

o Object name 

Default: *. 

e Extension name 

Default: “.GRF”. 

 

Description Similar to ^IL but you can move the image to any position 

by placing a ^FO command. 

  

Example 

 

^XA^CFD 

^FO50,20^AB^FDORIGINAL DATA0^FS 

^FO50,60^AB^FDORIGINAL DATA1^FS 

^ISR:SAM.GRF,N 

^XZ 

 

^XA^FO50,10^IMR:SAM.GRF^FS 

^FO50,100^AB^FDNEW DATA^FS 

^XZ 

Output 

^IS Saves image 

 

Syntax  ^ISd:o.e,p 

 

Parameter 

d Device driver name 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

o Object name 

Default: *. 

e Extension name 

Default: “.GRF”. 

p Print image during storing 

Default: Y for yes. 

Other: N for no. 

 

Description This command saves the current label pattern to a graphic 

file for future use. Refer to ^IL and ^IM for details. 

~JA Cancels All 

 

Syntax  ~JA 

 

Description This command cancels all format commands in the buffer 

and batches that are printing. 

 

After receiving this command, the printer stops after the 

current label is finished printing. All forms, graphics and 

soft fonts stored in printer will be erased. 

 

Example 

 

~JA 

^JB Initializes the flash memory 

 

Syntax  ^JBa 

 

Parameter 

a Device driver name 

Default: must be specified 

Others: E for Internal Flash. 

 

Description This command (^JBE) initializes the flash memory. After 

initialization all data in the flash memory will be cleared. 

~JC Media Sensor Calibration 

 

Syntax  ~JC 

 

Description The printer will feed some labels to measure the label 

length and recalibrate the media sensor. 

 

Example 

 

~JC 

~JD Enters dump mode 

 

Syntax  ~JD 

 

Description After receiving this command, the printer will enter the 

dump mode. In dump mode, all contents of the file will be 

printed with the HEX value below.  
 

Example 

 

~JD 

~JK Delay cut 

 

Syntax  ~JK 

 

Description When the printer is in the Delayed Cut Print Mode (Refer 

to command: ^MM), the printer will not cut any label until 

the printer receives” ~JK”.  

 

Example 

 

^XA 

^MMD 

^FO50,50^A9,30,35^FDD&S^FS 

^PQ3 

^XZ 

 

^XA~JK^XZ 

~JL Sets the label length 

 

Syntax  ~JL 

 

Description This command can set the label length. The label length 

depends on the label size. 

~JP Pause 

 

Syntax  ~JP 

 

Description After receiving this command, the printer enters pause 

mode. Pause occurs between labels, pressing the FEED 

button will have the next label to be printed. 

 

After receiving this command, the printer enters pause 

mode and clears the next format that would print. The 

function of this command is same as to use CANCEL on the 

printer.  

~JR Power on reset 

 

Syntax  ~JR 

 

Description This function acts same as cold start. All temporary data 

will be lost. 

^JS Sensor Selection 

 

Syntax  ^JSa 

 

Parameter 

a Sensor Selection 

Default: A for auto select 

Other: R for reflective sensor and T for see-through sensor 

^JZ Recovery control 

 

Syntax  ^JZa 

 

Parameter 

a Reprint 

Default: Y for reprint. 

Other: N for no recovery. 

 

Description Once the recovery is enabled after media-out or 

ribbon-out happens the printer will print the last label 

after error is removed. 

 

This parameter will be saved to permanent E2PROM or 

flash. 

^KD Select date and time format (for Real-Time Clock) 

 

Syntax  ^KDa 

 

Parameter 

a Date and time format 

Default: 0 for normal 

Other:  

1 for MM/DD/YY (24-hour clock) 

2 for MM/DD/YY (12-hour clock) 

3 for DD/MM/YY (24-hour clock) 

4 for DD/MM/YY (12-hour clock) 

 

Description This command can set the format of the RTC’s date and 

time.  

^LH Defines the new origin 

 

Syntax  ^LHx,y 

 

Parameter 

x Horizontal coordinate 

Default: 0. 

Others: 1 ~ max printing width of the printer. 

y Vertical coordinate 

Default: 0. 

Others: 1 ~ max printing length of the printer. 

 

Description All ^FO and ^FT commands are relative to the origin. The 

^LH defines the new origin. For the printer with left 

justification will ignore the x parameter. 

 

Example 

 

^XA 

^LH20,50^FO50,50^A0,30,35^FD(50,50) Location^FS 

^XZ 

^LL Defines the label length 

 

Syntax  ^LLa 

 

Parameter 

a Length in pixels 

 

Description The command is necessary when using the continuous 

media otherwise the printer does not know the actual 

length. 

 

Example 

 

^XA^MNN^LL300 

^FO50,50^AF^FDLength=300 pixels^FS 

^XZ 

 

 



 
 

163 
 

^LR Reverse print 

 

Syntax  ^LRa 

 

Parameter 

a Reverse 

Default: N. 

Other: Y for reverse print. 

 

Description The command is valid for the whole and subsequent 

labels. 

 

Example 

 

^XA^LRY 

^FO40,45^GB0,120,150^FS 

^FO80,60^A0,30,25^FDLR COMMAND^FS 

^FO120,120^A0,40,30^FDREVERSE^FS 

^XZ 

 

Output 

^LS Shifts the image 

 

Syntax  ^LSa 

 

Parameter 

a Shift left value in pixels 

Default: 0. 

Others: -999 ~ 999. 

 

Description The parameter is saved in the permanent memory. Since 

there are incompatibilities between printers with 

centralization and left justification, this command can 

compensate the horizontal position variances. 

 

Example 

 

^XA^LS100^XZ 

^LT Shifts the image vertically 

 

Syntax  ^LTa 

 

Parameter 

a Shift down value in pixels 

Default: 0. 

Others: -120 ~ 120.  

 

Example 

 

^XA^LT100^XZ 

^MC Clears the image frame buffer 

 

Syntax  ^MCa 

 

Parameter 

a Clear 

Default: Y. 

Others: N for keeping the image. 

 

Example 

 

^XA 

^FO80,60^A0,30,25^FDDATA 0^FS 

^MCN 

^XZ 

 

^XA 

^FO80,100^A0,30,25^FDDATA 1^FS 

^MCY 

^XZ 

 

Output 

^MD Sets the darkness 

 

Description  ^MDa 

 

Parameter 

a Darkness 

Default: 0. 

Others: -30 ~ 30. 

 

Description The parameters will not be saved to the permanent 

memory. 

 

This command adjusts the darkness relative to the current 

darkness setting. 

 

Example 

 

^XA^MD10^XZ 

^ML Sets maximum label length 

 

Syntax  ^MLa 

 

Parameter 

a Length in pixels. 

^MM Print mode 

 

Syntax  ^MMa 

 

Parameter 

a Default: T (tear off) 

Others: P for peel off mode, C for cut mode and D for delay 

cut mode (~JK). 

 

Description In cut mode, it cuts every label during printing. In delay 

cut mode, the printer will enter pause mode after each 

label is printed out; user must send ~JK command to let 

the cutter cut label.  

^MN Media type 

 

Syntax  ^MNa 

 

Parameter 

a Type 

Default: Y for die-cut or non-continuous media. 

Other: N for continuous media. 

 

Example 

 

^XA^MNN^XZ 

^MT Transfer mode 

 

Syntax  ^MTa 

 

Parameter 

a Mode 

Default: T for thermal transfer. 

Other: D for direct thermal. 

 

Example 

 

^XA^MTT^XZ 

 

This example sets thermal transfer mode and enables ribbon-out check. 

^MU Sets units 

 

Syntax  ^MUa 

 

Parameter 

a Unit 

Default: D for pixels. 

Others: I for inches and M for millimeters. 

 

Description Once the unit is changed, the commands with length 

parameters will be affected. 

 
 



 
 

173 
 

^PH Feeds a blank label after printing 

 

Syntax  ^PH  or  ~PH 

 

Description The ^PH command causes the printer to feed a blank label 

after printing the format. 

 

The ~PH command causes the printer to feed a blank label 

before printing the format or when the printer is in pause 

mode. 

^PM Mirror image 

 

Syntax  ^PMa 

 

Parameter 

a Direction 

Default: N for normal. 

Other: Y for reverse printing. 

 

Example 

 

^XA^PMY 

^FO620,50^AF^FDREVSERSE^FS 

^FO620,110^AF^FDDIRECTION^FS 

^XZ 

 

Output 

^PO Print orientation 

 

Syntax  ^POa 

 

Parameter 

a Direction 

Default: N for normal. 

Other: I for inverting the label 180。. 

 

Example 

 

^XA^POI 

^FO640,50^AF^FDINVERTED^FS 

^FO640,110^AF^FDFORMAT^FS 

^XZ 

 

Output 

^PP Programmable pause 

 

Syntax  ^PP or ~PP 

 

Description ^PP causes the printer to enter pause mode after the first 

label is printed. Pressing the FEED button from panel 

makes the printer to enter normal operation. 

 

~PP can make the printer enter pause mode at first. 

Pressing the FEED button from panel makes the printer to 

print normally. 

 

Example 

 

^XA 

^PP 

^FO0,0^A9,20,20^FDARGOX^FS 

^PQ2^FS 

^XZ 

 

^XA 

^FO0,0^A9^FDABCDE^FS 

^PQ3^FS 

^XZ 

^PQ Print control 

 

Syntax  ^PQq,p,r,o 

 

Parameter 

q Total quantity of labels 

Default: 1. 

Others: 2 ~ 32767. 

p Pause count and cut count 

Default: 0 for no pause and no cut   

Others: 1 ~ 32767. 

r Copies for each set 

Default: 0 for single copy. 

Others: 1 ~ 32767. 

o Override pause count 

Default: N. 

Other: Y. 

 

Example 

 

^XA 

^FO130,65^GB100,100,3^FS 

^FO140,79^A0,40,35^FR^SN1000,13,N^FS 

^PQ4,0,2 

^XZ 

 

Output 

^PR Print rate 

 

Syntax  ^PRa,b,c 

 

Parameter 

a Print speed 

Default: By model 

Others: 2 ~ 8 IPS. 

b Feed Speed  

(Ignored) 

c Backfeed speed  

(Ignored) 

 

Description The maximum speed depends on the printer models. 

 

Example 

 

^XA^MUD^LH0,0^MNY^LL609^PW812 

^PR4~TA0~SD0^MTT^MMT 

^FO77,79^GB686,438,438^FS 

^PQ1,1,0,Y^FS 

^XZ 

 
 
Note: Refer to 6.DENSITY AND SPEED. 

~PS Print start 

 

Syntax  ~PS 

 

Description The operation is identical to pressing the PAUSE or FEED 

button on the front panel. It makes the printer to continue 

printing after pause. 

 

Example 

 

^XA~PS^XZ 

^PW Print width 

 

Syntax  ^PWa 

 

Parameter Label width in dots. 

^SC Sets the communication protocol for the RS232 

 

Syntax  ^SCa,b,c,d,e 

 

Parameter 

a Baud rate 

1200 ~ 115200. 

b Data length 

7or 8 data bits. 

c Parity 

Always N for none parity. 

d Stop bits 

1 or 2 stop bits. 

e Handshake 

X for Xon/Xoff and D for hardware. 

 

Description The parameters will be saved to the permanent memory. 

 

Example 

 

^XA^SC9600^XZ 

 

This example sets the baud rate 9600. 

~SD Sets the darkness base 

 

Syntax  ~SDa 

 

Parameter 

a Base darkness 

Default: By model 

Others: 1 ~ 30. 
 

Description The parameters will be saved to the permanent memory. 

 

The setting can compensate the darkness without 

changing your application software after the printer is 

used for a long time. 

 

 
Note: Refer to 6.DENSITY AND SPEED. 

^SE Sets encoding table 

 

Syntax  ^SEd:f 

 

Parameter 

d Device driver 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

f Table name 

8 character file name with 3 characters extension name. 

The default extension name for table is “.DAT”. 

^SF Sets the serialization field 

 

Syntax  ^SFa,b 

 

Parameter 

a Mask string 

The string defines the substring of ^FD string to be serialized. 

The mask string starting with the right-most position. 

Mask string definition: 

D or d       Decimal numeric 0 ~ 9. 

H or h       Hexadecimal 0 ~ 9 plus A ~ F or a ~ f. 

O or o       Octal 0 ~ 7. 

A or a       Alphabetic a ~ z or A ~ Z. 

N or n       Alphanumeric 0 ~ 9 plus a ~ z or A ~ Z. 

%          Skip the serialization for current character. 

b Serialization string 

The string is used to add the field on each label. 

 

Example 

 

^XA 

^FO20,10^AF^FDFixed Data^FS 

^FO20,60^AF^FDBL00-0^SFAAdd%d,1%1^FS 

^PQ2 

^XZ 

 

Output 

^SL Sets mode (for Real-Time Clock) 

 

Syntax  ^SLa 

 

Parameter 

a Time mode 

Default values: S for start time mode  

(The time can read from RTC when label formatting with 

^XA. The first label will be the same with the last label.) 

 

Other Values:  

T for time now mode 

(The time can read from RTC when label to be printed in 

queue. Each printed time is different.) 

 

Description This command can define the RTC’s mode of operation.  

^SN Sets counter 

 

Syntax  ^SNa,b,c 

 

Parameter 

a Start value 

Default: 1. 

Others: 12-digit maximum. 

b Step value 

Default: 1. 

Others: 12-digit maximum. 

c Leading zeros 

Default: N for no leading zeros. 

Other: Y for that the leading zeros exist. 
 

Example 

 

^XA 

^FO80,30^GB200,150,3^FS 

^FO100,40^A0,40,35^SN1000,13,N^FS 

^FO100,90^AF^SN555,1^FS 

^PQ2 

^XZ 

 

Output 

^SO Sets offset (for Real-Time Clock) 

 

Syntax  ^SOa,b,c,d,e,f,g 

 

Parameter 

a Clock set 

Default: Value must be defined 

Others:2 for secondary, 3 for third 

b Months offset 

Default: 0 

Others: -32000~3200 

c Days offset 

Default: 0 

Others: -32000~3200 

d Years offset 

Default: 0 

Others: -32000~3200 

e Hours offset 

Default: 0 

Others: -32000~3200 

f Minutes offset 

Default: 0 

Others: -32000~3200 

g Seconds offset 

Default: 0 

Others: -32000~3200 

Example 

 

^XA 

^SO2,-1,10,-5,20,-10,0 

^SO3,1,-10,5,-20,10,0 

^XZ 

 

 

^XA 

^FO20,100^A0N,50,50 

^FC$,!,% 

^FDPrimary: $b/$d/$Y/$H/$M^FS 

^FO20,200^A0N,50,50 

^FC$,!,% 

^FDSecondary: !b/!d/!Y/!H/!M^FS 

^FO20,300^A0N,50,50 

^FC$,!,% 

^FDThird: %b/%d/%Y/%H/%M^FS 

^FO20,400^A0N,50,50 

^XZ 

 

Output 

^ST Sets date and time (for Real-Time Clock) 

 

Syntax  ^STa,b,c,d,e,f,g 

 

Parameter 

a Month 

Default: None 

Others:01~12 

b Day 

Default: None 

Others: 01~31 

c Year 

Default: None 

Other: 0000~9999 

d Hour 

Default: None 

Other: 00~23 

e Minute 

Default: None 

Other: 00~59 

f Second 

Default: None 

Other: 00~59 

g Format 

Default: M 

Other: A for a.m., P for p.m., M for 24-hour military 

~TA Sets tear-off adjust position 

 

Syntax  ~TAa 

 

Parameter 

a Adjustment in pixels 

Default: 0. 

Others: -64 ~ 64. 

 

Description This command lets you adjust the reset position of the 

media after a label is printed. 

^TB Text blocks 

 

Syntax  ^Tba,b,c 

 

Parameter 

a Block Orientation 

Default: N (Portrait) 

Others: I for reverse portrait, B for landscape and R for 

reverse landscape. 

B Block width in dots 

Default: 1 dot 

c Block height in dots 

Default: 1 dot 

 

Description This command can print a text block with defined width 

and height. This command has an automatic word-wrap 

function. 

If the text is over the block height, the text is truncated. 

This command supports complex text layout features. 

 

Example 1 

 

^XA 

^CF0,30,30^FO120,200 

^PR4,4,4 

^TBN,100,500 

^A0N,50,50^FD ARGOX PRINTER GOOD! Apple 12345 ABCDEFG. 

TBN&A0N^FS 

^XZ 

 

Example 2 

 

^XA 

^CF0,30,30^FO120,200 

^PR4,4,4 

^TBR,100,500 

^A0R,50,50^FD ARGOX PRINTER GOOD! Apple 12345 ABCDEFG. 

TBR&A0R^FS 

^XZ 

 

Example 3 

 

^XA 

^CF0,30,30^FO120,200 

^PR4,4,4 

^TBI,100,500 

^A0I,50,50^FD ARGOX PRINTER GOOD! Apple 12345 ABCDEFG. 

TBI&A0I^FS 

^XZ 

 

Example 4 

 

^XA 

^CF0,30,30^FO120,200 

^PR4,4,4 

^TBB,100,500 

^A0B,50,50^FD ARGOX PRINTER GOOD! Apple 12345 ABCDEFG. 

TBB&A0B^FS 

^XZ 

 



 
 

197 
 

^TO Copy objects 

 

Syntax  ^Tod:o.e,s:o.e 

 

Parameter 

d Device driver name 

s 
Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

O Object name 

e Extension name 

 

Description This command is used to copy an object to another object 

or other device. Before you copy the object to other 

device, make sure that the device exists. 

 

Example 

 

^XA^TOABC.GRF,B:XXX.GRF^XZ 

 

 



 
 

198 
 

~WC Prints the configuration list 

 

Syntax  ~WC 

 

Description Data in this list include the followings: 

 

Font List 

    Symbol set 

Firmware  

Version and date code 

Checksum data 

Hardware 

    Memory size 

    Accessories 

    RS232 protocol 

    Media sensor 

Media 

    Media type 

    Ribbon mode 

    Print length 

Settings 

    Control code definition 

    Reprint control 

^WD Prints the directory on a label 

 

Syntax  ^WDd:o.e 

 

Parameter 

d 

Device driver name 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

O Object name 

e Extension name 

 

Example 
 

^XA^WD^XZ 

 

Output 

^XA Starts format 

 

Syntax  ^XA 

 

Description The ^XA command is the beginning bracket. It indicates 

the start of a new format. 

^XB Disables backfeed 

 

Syntax  ^XB 

 

Description This command disables tear-off to improve the 

throughput. 

^XF Retrieves form 

 

Syntax  ^XFd:o.e 

 

Parameter 

d Device driver name 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

O Form name 

e Extension name 

Default: “.PPLZ”. 

 

Example 

 

^XA^DFA.PPLZ^FS 

^FO40,40^GB150,158,3^FS 

^BY2,3,40 

^FO50,60^A0,30,28^FN1^FS 

^FO50,110^B2^FN2^FS 

^XZ 

 

^XA^XFA.PPLZ^FS 

^FN1^FDFORM EXEC^FS 

^FN2^FD2468^FS 

^XZ 

 

Output 

^XG Retrieves graphic 

 

Syntax  ^XGd:o.e,x,y 

 

Parameter 

d 

 

Device driver name 

Default: R (RAM) 

Others: B (USB memory), E (Flash memory) 

O Graphic name 

e Extension name 

Default: “.GRF”. 

x Scale factor along horizontal axis 

Default: 1. 

Others: 2 ~ 10. 

Y Scale factor along vertical axis 

Default: 1. 

Others: 2 ~ . 

 

Example 

 

~DGA.GRF,42,6, 

00000FF00000 

00000FFFF000 

FFFFFFFFFFF0 

FFFFFFFFFFFF 

FFFFFFFFFFF0 

00000FFFF000 

00000FF00000 

 

 

^XA 

^FO30,40^XGA.GRF,1,1^FS 

^FO40,60^XGA.GRF,5,5^FS 

^FO30,100^XGA.GRF,2,3^FS 

^XZ 

 

Output 

^XZ End format 

 

Syntax  ^XZ 

 

Description The ^XZ command is the ending bracket. The format starts 

with ^XA and ends by ^XZ. 

6.DENSITY AND SPEED 

 

PPLZ 
Command 

Syntax 

WLP-660 

WLP-666 

WP-660 

WP-666 

WP-880 

WP-888 

WL-880 

WL-888 
I4-240 I4-250 I4-350 

Density 

range 
~SDa 

1~30 1~30 1~30 1~30 1~30 1~30 1~30 

Default 

density 
8 15 15 15 16 16 16 

Speed 

range ^PRa 

(Unit: ips) 

2~6 2~7 2~6 2~6 2~6 2~8 2~6 

Default 

speed 
5 5 3 3 5 6 5 

7.COMMAND REFERENCE CHART 

 

COMMANDS DESCRIPTION 
WLP 

series 

WL series 

WP series 

I4 

series 
 

^A Font Selection * * *  

^A@ Font Selection by the font name * * *  

^B0 Bar code – Aztec * * *  

^B1 Bar code – code 11 * * *  

^B2 Bar code – Interleaved 2 of 5 * * *  

^B3 Bar code – Code 39 * * *  

^B5 Bar code – Planet * * *  

^B7 Bar code – PDF-417 * * *  

^B8 Bar code – EAN-8 * * *  

^B9 Bar code – UPC E * * *  

^BA Bar code – Code 93 * * *  

^BC Bar code – Code 128 with subsets A, 

B and C 

* * *  

^BD Bar code – Maxicode * * *  

^BE Bar code – EAN-13 * * *  

^BF Bar code – Micro PDF417 * * *  

^BI Bar code – Industrial 2 of 5 * * *  

^BJ Bar code – Standard 2 of 5 * * *  

^BK Bar code – Codabar * * *  

^BL Bar code – Logmars * * *  

^BM Bar code – MSI * * *  

^BO Bar code – Aztec * * *  

^BP Bar code – Plessey * * *  

^BQ Bar code – QR code * * *  

^BR RSS Bar Code (GS1 DataBar) * * *  

^BS Bar code – UPC/EAN extension * * *  

^BT Bar code – TLC39 * * *  

^BU Bar code – UPC A * * *  

^BX Bar code – Data Matrix  * * *  

^BY Parameter defaults for general bar 

codes 

* * *  

^BZ Bar code – Postal * * *  

^CC Changes the caret character * * *  

^CD Changes the delimiter character * * *  

^CF Changes default font * * *  

^CI Changes symbol set * * *  

COMMANDS DESCRIPTION 
WLP 

series 

WL series 

WP series 

I4 

series 
 

^CT Changes the tilde character * * *  

^CW Defines the font ID * * *  

~DB Downloads a bitmap font * * *  

~DE Downloads encoding * * *  

^DF Downloads form * * *  

~DG Downloads graphic * * *  

~DN Abort Download Graphic * * *  

~DU Downloads unbounded TrueType 

font 

* * *  

~EF Erases all stored forms * * *  

~EG Erases all stored graphics * * *  

^FA Field allocate * * *  

^FB Defines the format of a block data * * *  

^FC Field Clock * * *  

^FD Field data * * *  

^FH Field HEX * * *  

^FN Field number * * *  

^FO Moves the position for print pattern * * *  

^FP Field Parameter * * *  

^FR Reverse print * * *  

^FS End of field * * *  

^FT Defines coordinate for print pattern * * *  

^FV Defines variable * * *  

^FW Defines default orientation * * *  

^FX Comment * * *  

^GB Box or line * * *  

^GD Diagonal line * * *  

^GF Direct graphic * * *  

^GS Prints special symbols * * *  

^HF Host Format * * *  

^HH Configuration Label Return  * *  

~HI Host identification  * *  

~HM Memory status  * *  

~HS Host status  * *  

^HW Directory list  * *  

^ID Erases graphics * * *  

^IL Load stored images * * *  

^IM Moves image * * *  

^IS Saves image * * *  

COMMANDS DESCRIPTION 
WLP 

series 

WL series 

WP series 

I4 

series 
 

~JA Cancels All * * *  

^JB Initializes the flash memory * * *  

~JC Media Sensor Calibration * * *  

~JD Enters dump mode * * *  

~JK Delay cut  * *  

~JL Sets the label length * * *  

~JP Pause * * *  

~JR Power on reset * * *  

^JS Sensor Selection * * *  

^JZ Recovery control * * *  

^KD Select date and time format (for 

Real-Time Clock) 

* * *  

^LH Defines the new origin * * *  

^LL Defines the label length * * *  

^LR Reverse print * * *  

^LS Shifts the image * * *  

^LT Shifts the image vertically * * *  

^MC Clears the image frame buffer * * *  

^MD Sets the darkness * * *  

^ML Sets maximum label length * * *  

^MM Print mode * * *  

^MN Media type * * *  

^MT Transfer mode  * *  

^MU Sets units * * *  

^PH Feeds a blank label after printing * * *  

^PM Mirror image * * *  

^PO Print orientation * * *  

^PP Programmable pause * * *  

^PQ Print control * * *  

^PR Print rate * * *  

~PS Print start * * *  

^PW Print width * * *  

^SC Sets the communication protocol for 

the RS232 

 * *  

~SD Sets the darkness base * * *  

^SE Sets encoding table * * *  

^SF Sets the serialization field * * *  

^SL Sets mode (for Real-Time Clock) * * *  

^SN Sets counter * * *  

COMMANDS DESCRIPTION 
WLP 

series 

WL series 

WP series 

I4 

series 
 

^SO Sets offset (for Real-Time Clock) * * *  

^ST Sets date and time (for Real-Time 

Clock) 

* * *  

~TA Sets tear-off adjust position * * *  

^TB Text blocks * * *  

^TO Copy objects * * *  

~WC Prints the configuration list * * *  

^WD Prints the directory on a label * * *  

^XA Starts format * * *  

^XB Disables backfeed * * *  

^XF Retrieves form * * *  

^XG Retrieves graphic * * *  

^XZ End format * * *  

